First Person Shooter game
This tutorial explains the implementation in Flash of a game in which bombs drop randomly from the sky and the player moves and fires an anti-aircraft gun using the arrow keys.  The game keeps track of successful hits, misses, and bombs landing.  The project is intended as a base or reference for you to create your own game.

A very important part of computer programming is dividing big tasks into small tasks.  Learning how to do this takes experience.  This tutorial divides the project into 11 stages.  You can and probably should read or skim the whole tutorial before opening up Flash.  However, then proceed slowly, making sure each project works before proceeding to the next.  The mottos to keep in mind are "divide and conquer" and "make haste slowly".

Description of game

The game in process is shown in this screen shot:

The yellow circle is the gun (you can certainly change this to be more gun-like: see a screen shot at the end of this tutorial for an improved graphic) and the maroon circles are the bombs. The large number is a result of the time taken to take the screen shot.  The tan splotch represents an exploded bomb and the orange a bomb in the last stages of exploding.  The text fields on the upper right show the counts that are accumulated by the system.  The player moves the gun to the left and right using the arrow keys.  The gun is fired using the up arrow key.  Bombs that are hit in the air disappear as does the fired missile (referred to as anti-aircraft in this tutorial).

Critical design features

The project as defines requires the following features.  They range from the general to the specific.  It makes sense to read this and then re-read it after completing the project.
· You need to program the behavior of the bombs, the gun, and the antiaircraft (aa) independently.  This is achieved using what is called in computer science object oriented, event driven programming.  Specifically, the code for responding to events for any one bomb, the single gun, and any aa is defined by itself, independently of anything else, in the actions for these objects.  This includes making the bombs fall down (move down the screen) and the aa travel up (move up the screen).  Flash itself, that is, the Flash system or engine, does the rest.
· Your code must produce multiple bombs and multiple aas.  This is done by creating bomb and aaseed symbols, placing in the timeline but off stage and writing the event handling code.  Then other code (in frames for the bombs and in the event handling for keys) uses the duplicateMovieClip method to make duplicates.  

· The position and timing of the bombs must appear random.  This is done using the Math.random() method.  The code invoking the random method occurs as ActionScript in a frame.  The first frame contains variables and function definitions.  The second frame (labeled "continue") contains the code for dropping bombs.  The code in the third frame is gotoAndPlay("continue").
· The game must respond to the player's moves.  This is done using onClipEvent(keyDown) associated with the gun instance named gunner. Note: it could actually be associated with any object.  This event handler calls a function (keytest) that includes code making use of the Key object, such as
 if (Key.isDown(Key.RIGHT)) {gunner._x += unit;}
· The code must detect and react to a bomb hitting the ground and an aa hitting a bomb.  This is done using the hittest method of a movie clip.
Construction Plan
At the risk of being repetitious, you may try to do this project without closely following this tutorial, but do work incrementally: code and test then add and test again. The project was built starting with fps1 and proceeding to fps10, building on the prior project.  Keep in mind that when you change a symbol in the Library, all instances are changed.  If you create a new keyframe, all material is copied over from the last keyframe. However, if you add or modify a keyframe after creating successive frames, if you want the change to be present in the successive frames, you need to use copy and paste in place to make the change to the other frames.
fps1: Create a gun that moves to the left and to the right in response to the player pressing the arrow keys. 

The first step for this and most of my other projects is to create two layers on the main timeline: one named board and the other names actions.  I put all of the frame code in the actions layer and all the graphic material in the board layer.  Furthermore, I make heavy use of functions and put the function definitions in the actions layer, first frame.  The object action code frequently calls the functions.  This is not required but is a practice that has served me well.  

Use Insert/New Symbol and name the new symbol gun and make it a movie clip.  Center a graphic (I used the oval drawing tool) on the origin. Go back to Scene 1 (the main timeline) and move an instance of this symbol to the stage (frame 1, board layer).  Name the instance gunner. Making sure that gunner is selected, open up the Actions panel. You may need to click on the upper right arrow and select Expert mode. While that panel is open, click on View line numbers.  This becomes useful for interpreting messages about syntax errors.  In the Actions panel, write the following:
onClipEvent(keyDown) {


_root.keytest();

}

What this means is: if and when any key is pressed down, call the function keytest defined in the _root (the main) timeline.  You will next define the keytest function.

Click on frame 1, actions layer, and put in the following code:

var unit = 10;

function keytest() {


if (Key.isDown(Key.RIGHT)) {

    gunner._x += unit;

       }


if (Key.isDown(Key.LEFT)) {

     gunner._x -= unit;

   }

}

The variable unit acts essentially as a constant in this project. It indicates the number of pixels that the gunner moves at each key press. You can change the value from 10.  The function keytest performs two independent tests. The code can be read as follows: if the key that is down is the right arrow key, then change the current x (horizontal) position of gunner by adding unit to it.  If the key that is down is the left arrow key, then change the current x position by subtracting unit from it.  In subsequent steps, we will add the test for the UP key.
Go to Control/Test Movie and see what happens when you press the arrow keys.

fps2: Create a bomb that falls down the screen.

Go to Insert/New Symbol and create a movie clip named bomb. Mine was again a simple circle. I drew a circle and then repositioned it by selecting both the fill (inside) and the stroke (outside) and using the Info panel, set the center to be x=0 and y=0. Go back to Scene 1 and move an instance (name it bomb) to the board layer, frame 1.  Place it high (near the top) of the stage. While the bomb instance is selected, write in the Actions panel:
onClipEvent(enterFrame) {


this._y += _root.bspeed;

}
Assume (this means: have faith that you will do it) that bspeed is a variable defined in the actions layer, frame 1.  The code for the bomb means: on every frame of the bomb movie clip (yes, there is just one frame, but it is played and re-played since all movie clips loop unless instructed otherwise by ActionScript), change the _y (vertical) position of this object (that is, the bomb) by adding the value of the bspeed variable.  The this is a keyword and refers to the instance for which the code is action code.  

Click on the actions layer and in the actions panel, add the following line that sets up bspeed as a variable and sets the value.  You can do this right after the var statement for unit:

var bspeed = 3;
You can now test the project.  The bomb will start to fall immediately and nothing stops it except closing the project. However, before doing that, test (re-test) that the arrow keys still move gunner back and forth. Always check to make sure 'old stuff' still works after adding new stuff!

fps3: Add ground (called homeland) and make the bomb stop when it hits the homeland

Use Insert/New Symbol to add a movie clip named homeland.  I created a simple, long rectangle.  Move an instance of this symbol to the board layer, first (and only so far) frame and position it at the bottom of the stage. Name it homeland. Select the bomb. The Action panel should be showing the code created for fps2.  Let's think about what we want this code to be.  It may seem convoluted, but [my] answer is the following: if this bomb is not hitting the homeland, move it down.  The code, therefore, is the following: the previous one-line is now the body of an if statement.  The if condition is read as: apply the hittest method for this instance (the bomb) against the homeland instance on the _root (main) timeline.  If it is NOT true, as indicated by using the ! to indicate negation), then do the clause.
onClipEvent(enterFrame) {


if (!(this.hittest(_root.homeland))) {


     this._y += _root.bspeed;}

}
Try the project.  Do test that the arrow keys work. The bomb should settle on the homeland.

At some LATER time, you may want to make the home land more interesting.  Do keep in mind that the hittest method is in terms of bounding rectangles. If you want a more subtle interaction, you 

fp4: Add the firing of antiaircraft (aa) emanating from gunner.

The next step requires a new symbol, named aaseed, which will be duplicated. Use Insert/New Symbol, and create a new movie clip named aaseed.  I made it an elongated oval, centered at 0, 0.  Move an instance of this off stage, to the board layer, frame 1.  Name the instance aaseed.  In the next step, we will add object action. For now, click on the actions layer.  Before getting into the details of the code, here is background on what is to be done.  The player can shoot off a lot of aa's.  It will be necessary to keep track of these to remove them.  For this reason, your code will create both a duplicate instance of the aaseed and also create an object that points to that instance.  All current such instances will be stored in an array.  

Add two var statements:

var nl = 1;

var aalist = new Array();

The variable nl is necessary to keep track of the levels of the instances created by duplicating movie clips.  There must be just one instance per level.  The aalist variable is the array described above.

Add to the keytest function, the if test and the action when the player presses the up arrow.


if (Key.isDown(Key.UP)) {



_root.fire();


}

As you can see, the action take if the if condition is met is to call a function named fire which, in turn, calls a function named Makeaa.
function fire() {


var aa = new Makeaa();


aalist.push(aa);  

}

The push method (of the array aalist) adds the newly created aa object to the array.

My name (and I say that to emphasize it was my choice) starts with a capital because it is a convention (not required, but done as a standard practice) to distinguish functions that create objects.  Please note that two creations happen here: the duplication of the aaseed movie clip and the definition of this, the thing returned by the Makeaa function.  What this is is a pointer to the new instance.

function Makeaa() {


var aan = "aa"+nl;


aaseed.duplicateMovieClip(aan,nl++);


_root[aan]._x = gunner._x;


_root[aan]._y = gunner._y-gunner._height;


this.aa = _root[aan];

}

What this code does is make what is produced by the code, the thing indicated by the special built-in term this have a property named aa.  The property named aa is the latest (aan is its instance name) duplicate of aaseed created.  Notice that the new aa's will be named aa0, aa1, aa2, and so on.  Notice also that it is positioned to be right above gunner.  These are all pushed onto the array aalist in the function fire described previously. 
The aa does not go up into the virtual sky.  Test the project.  You may be able to see the 'off-stage'  aaseed.  The bomb will come down. You can move gunner back and forth.  If and when you press the up arrow, something will appear right above gunner.  The screen shot shows one possibility:

fps5: Make the antiaircraft fire (the aa's) go up into the virtual sky and stop at the top.
This step is actually pretty easy given what you have done so far.  In fact, it accomplishes what we took two steps to do for the bomb:  we will make the antiaircraft move up AND make it stop.  Select the aaseed instance and write the following in the Action panel:

onClipEvent(enterframe) {


if (this._y>0) {


this._y -= _root.aaspeed;


}

}
What this code does is check if the vertical position is greater than zero, meaning that it is not at or above the top.  If this is the case, then re-position the object by subtracting the value of a variable named aaspeed defined in the _root (main) timeline.  You need to click on the actions layer, frame 1, and add the var statement for aaspeed:

var aaspeed = 10;

Now the magic that happens here (and the value of object oriented event driven programming), is that the code defined for the original instance, the one named aaseed, is copied for all the duplicated instances.  Each aa instance independently moves up the screen.  Try it out.  Move gunner back and forth. Press on the up arrow. You should end up with something like this.

fps6: Check if any antiaircraft fire (aa's) hit the bomb.
It took me a few steps to accomplish the task of making the aa's knock out a bomb.  In this step, I use the trace function to indicate a hit.  Add the following function in the actions layer, frame 1:

function checkforaa(thisbomb) {


var i;


var naa = aalist.length;


for (i=0;i<naa; i++) {



if (thisbomb.hitTest(aalist[i].aa)) {



   trace("hit");




}


}
The function is called with a parameter indicating which bomb. At the moment, there is just one bomb, but that will change. The code uses a for loop to check all the aa's.  The hittest method of thisbomb is called against each aa1, aa2, etc. (each instance pointed to by the aa property of each element in aalist.  If there is a hit, then the trace function writes out the string "hit".  This function is called from the action code for the bomb, shown here in its entirety.  All you need to do is add the one line calling the checkforaa function defined in the main timeline with the parameter this.

onClipEvent(enterFrame) {


if (!(this.hittest(_root.homeland))) {


     this._y += _root.bspeed;}


_root.checkforaa(this);

}

I moved the single bomb right over the gunner instance.  Test the function.  You will need to fire (press the up arrow).  If you fire repeated and do 'hit' the bomb, you will see repetition of 'hit' in the trace panel.  Nothing else happens, however. Wait for the next step.
fps7: Make the aa and the bomb disappear when a hit takes place.

Making instances disappear is easy.  Replace the statement calling trace with:

thisbomb._visible=false;



aalist[i].aa._visible=false; 
The interpretation of the second line is: identify the ith element of the aalist array; identify its aa property (which will be a movie clip); make its _visible property false.

The code does not make the aa go away.  We will see the implications of this later. Test your project.

fps7a: Add the feature of multiple bombs, falling at random times and from random places. Remove any hit bombs (don't just make them invisible).
This is a major step in the implementation.  However, the coding does re-use many concepts already introduced.  We also did the prior steps in anticipation of multiple bombs.  Select the single bomb and make its action code be the following:
onClipEvent(enterFrame) {


if (!(this.hittest(_root.homeland))) {


     this._y += _root.bspeed;}


if (_root.checkforaa(this)) {



this.removeMovieClip();


}

}
What you are doing here is assuming (again, have the faith that you will get to it—you can't do everything at once) that the checkforaa function will return true if the bomb was hit and false otherwise, the code makes an if test and the true condition clause removes the instance.  The removeMovieClip method only works for movie clips created using duplicateMovieClip (or attachMovieClip, which we don’t cover), but that is the situation here.  Or, more correctly, it will be when we add the code to create new bombs.  This technicality of removeMovieClip is the reason why this step consists of the addition of multiple bombs and the removing of the bombs.
Now you need to change checkforaa, the new code is:

function checkforaa(thisbomb) {


var i;


var naa = aalist.length;


var nothit = true;


for (i=0;i<naa && nothit; i++) {



if (thisbomb.hitTest(aalist[i].aa)) {




nothit = false;



   aalist[i].aa._visible = false;



   
}


}


return (!nothit);

}
Compare it to the prior version. One change is that checkforaa does less of the work, but returns a value so that the calling code may do something.  The checkforaa function makes use of a variable I named nothit.  Notice that the aa instance is made invisible, but it still exists. The thisbomb instance is not made invisible.  However, on return from the function, it will be removed totally if it was hit.  The nothit variable starts off being true and will be set to false if and only if there is a true return from the hittest function.  The return statement returns the negation of nothit.  This could be done a variety of ways. You just need to be consistent.  It does make sense to make the 'English' of the variable names meaningful.
Now we need to add the code to drop bombs randomly from the sky.  For each layer, create two new key frames.  Give the second frame a label of "continue" (using the Property panel).  The new frames will have all the graphical content of the prior frames. However, if and when you add anything, to any frame, if it needs to be in other frames, you need to add it explicitly, probably using copy and paste in place.  Click on the actions layer, frame 1.   Add a new variable, called bombodds and set it to be .05:

var bombodds = .05;

Click on frame 2, actions layer, and add the following:

if (Math.random()<bombodds) {


bombn = "bomb"+nl;


bomb.duplicateMovieClip(bombn,nl++);


_root[bombn]._x = Math.random()*400 + 50;


_root[bombn]._y = Math.random()*100;

}

Recall from other projects that Math.random() returns a value from zero to 1.  If that value is below .05, which presumably happens roughly 1/20th of the time, the true clause of the if statement is executed. A variable (not declared, but this is okay) called bombn is set to be the string 'bomb' followed by the value of nl converted to a string. Notice we use the same value for the duplicated aa's and the bombs to keep the levels unique.  The code creates a new instance, using duplicateMovieClip as an instance of bomb.   To say this in the language of computer science: bomb and aaseed are each movie clip objects.  All movie clip objects have a method called duplicateMovieClip.  The next two lines call on Math.random() again to position the newly created instance, identified using the format of _root followed by the name of the instance in square brackets.  The positioning is from x = 50 to x = 450 and from y = 0 to y = 100.  
Click on the 3rd frame of the actions layer and put in the code:

gotoAndPlay("continue");

Test your project. You may notice two things.  If an aa misses, it does not disappear, but goes up to the top and stops, still visible. This will be changed in subsequent steps. Secondly, if two bombs fall in such a way that an aa hits both, then both bombs disappear.  I decided to keep this second effect.  
fps8: Keep track of the number of successful hits using text fields.
You must now add text fields to each frame.  In the board layer, create a static text field to use as a label and a dynamic text field var name hitcount.  Make sure that the text color makes the fields legible.  You also can change the justification to be to the right as is suitable for numbers.  Put in a zero (though this isn't necessary—the code does the right thing for an empty string).  Select both fields (keep the shift key down) and click on Edit/Copy. Then go to frame 2 and then frame 3 (all in the board layer), and do Edit/Paste in place.  

Now you need to think about where to put the code to increment the count.  The answer is in the if true clause for the condition _root.checkforaa(this):
if (_root.checkforaa(this)) {



_root.hitcount++;



this.removeMovieClip();


}

You need to change this in each frame. Test your project.  You should see the hit count go up if and when you/the player hits any bombs.

fps9: Keep track of the number of unsuccessful shots.
Add two more text fields (and you will need to copy and past them to the other frames) all board layer.  The static field holds 'wasted shots'.  The dynamic field is named wastecount: the var field in the property panel should hold wastecount.

Now again, you need to ask yourself the question: when can we (meaning the code) know that the shot is wasted, that is, has not been successful?

onClipEvent(enterframe) {


if (this._y>0) {



this._y -= _root.aaspeed;


}


else if (this._visible) {


_root.wastecount++;


this.removeMovieClip();


}

}

This code ups the wastecount field and also removes the movie clip.  It does make sense to remove movie clips you do not need, because the Flash engine spends resources on them.

This code also has to be changed in each frame. (Note: if we made the object code call a function defined in the actions layer, frame 1 of the main timeline, this would be easier because the object actions would not need to be changed.)

fps10: Keep track of the bombs that have landed and also make the landing more dramatic.

Add two new text fields to all frames, board layer: one static holding the words Bombs landed or some equivalent and one dynamic with var name bomblands.

Now is where you can make use of your drawing and dramatic skills. If the Library panel is not visible, go to Window and click on Library. Now double click on the bomb.  You will insert new keyframes and make several frames of the bomb exploding.  In the first frame (I didn't make a new layer: bad practice, but perhaps acceptable) and the last frame, insert in the Action panel: stop();  
The approach here is to keep the bombs in the first frame until (if and when) they hit the homeland.  The onClipEvent(enterframe) code now must distinguish between the first frame and subsequent frames.  This is done using the _currentframe property of movie clips.  Recall that your new bomb has multiple frames but initially is stopped at the first frame.   When the hittest indicates that this movie clip (I mean the this movie clip) has hit the homeland, the else clause in the code given next is taken.  Two things happen: the bomblands dynamic text field is incremented and the movie clip is 'told' to gotoAndPlay starting at frame 2.

onClipEvent(enterFrame) {


if (this._currentframe==1) {


if (!(this.hittest(_root.homeland))) {


     this._y += _root.bspeed;}


else {



_root.bomblands++;



this.gotoAndPlay(2);


}

 


if (_root.checkforaa(this)) {



_root.hitcount++;



this.removeMovieClip();


}


}

}
Notice that everything, including the if statement with the condition calling on checkforaa is within the clause for the currentframe being 1, so that nothing is done if the bomb is exploding or has exploded.
Try your project.

fps11: Inserting a start button and pressing the space key to stop; improved gun 
It turns out that the project as it now stands has problems in the published version relating to something called the focus.  To fix this problem and also make the game nicer by not beginning immediately, let us now add a button to start the game and use the space key to stop it.  In the first frame, actions layer, add:
stop();

This means that the movie will not go to the second frame and start dropping bombs unless something else happens.

To add the button, use Window/Common Library/Button. Pick any button you like and move it to the board layer, frame 1.  With the button instance still selected, add to the actions panel:

on (release) {


_root.gotoAndPlay("continue");

}
This starts the game going by going to the second frame where the bombs start dropping. It also has the effect of fixing the focus problem so key presses are recognized.  

Add a static text field to show how to play the game.

We also should add a graceful (or more graceful) way to stop the game.  Insert in both layers a new keyframe. Give this frame the label "done". (It does not matter which layer bears this label.)  In the actions layer, add 

stop();  
In the board layer, delete gunner, aaseed and bomb.  If you wish, you can move the static text fields over to the center and add a new one to say something about re-starting (see the game).  

In the actions layer, frame 1, add to the keytest function code to respond to the event of the player pressing the space bar:

if (Key.isDown(Key.SPACE)) {



_root.goToAndPlay("done");


}

This takes the _root movie to the done frame, stopping the game while still giving the player a chance to see the scores.  Any bombs in flight will still land.

You also need to make the aaseed and the bomb invisible in the continue frame.  Insert the code:

aaseed._visible = false;

bomb._visible = false;

The fact is that neither of these instances are visible in the published version of the project.  However, it is necessary to make aaseed invisible to prevent the missed shot count from incrementing.  I make the bomb invisible just to be consistent.

You certainly can make the gun (instance name gunner) something other than a circle.  Here is an improvement accomplished by 1) changing the gun symbol in the Library and 2) making the gunner instance lay on the ground.  Please note the following two factors: Changing the graphics of a symbol changes all instances.  Changing the location of the instance must be done in each frame.

Enhancements

There are clearly graphical enhancements to make to the project.  You can change the look of homeland, but keep in mind that the test is bounding rectangle versus bounding rectangle.  If you want a more exact test, you could make the homeland multiple clips and do a hittest against each.  Alternatively, you could determine the origin of the bomb, and use another form of the hittest method.  Assuming this refers to the bomb:

         _root.homeland(this._x, this._y, true);

This means the hit takes place only when the origin of the bomb intersects with any occupied pixel of the homeland movie clip.

At you can figure out by playing the game or examining the code, the bombs have no direct effect on the gun.  You may want to change this!

It also would be appropriate to have a scoring system that did not simply keep track of the 3 situations: bombs hit, shots missed and bombs landed.  For example, you could have the player start with a given amount of points (resources) and have that score go down by one small amount for every shot and by one large amount for every bomb landing.  A successful hit would add to the score.  You could add a time element: what is the score after so many seconds.

Other possibilities are to add different types of anti-aircraft (signed by pressing on different keys) and different types of bombs.
